DiffAlg: A DIFFERENTIAL ALGEBRA PACKAGE

MANUEL DUBINSKY, CÉSAR MASSRI, ARIEL MOLINUEVO,
AND FEDERICO QUALLBRUNN

ABSTRACT. In this article we present DiffAlg, a differential algebra package for Macaulay2. It can perform the following operations: wedge products and exterior differential of differential forms, contraction and Lie derivative of a differential form with respect to a vector field and Lie brackets between vector fields.

Given an homogeneous differential operator of degree one D, the lack of an algebraic module structure attached to the kernel or image of D hinders the study of D. The main purpose of DiffAlg is to handle these spaces degree-wise.

1. Motivation and description of the package

Algebraic and differential operations arise naturally when working with differential forms and vector fields, *e.g.*: wedge products and exterior differential of differential forms, contraction and Lie derivative of a differential form with respect to a vector field and Lie brackets between vector fields. Some important problems involving these operations include the following:

(a) A differential r-form ω in the affine space $\mathbb{K}^{n+1}$ *descends to the projective space* $\mathbb{P}^n_{\mathbb{K}}$ if it satisfies the equation
$$i_R\omega = 0,$$
where R is the radial vector field $R = \sum x_i \frac{\partial}{\partial x_i}$ and i_R denotes the contraction, see [3, Theorem 8.13, p. 176].

(b) If ω is a differential 1-form, then ω defines a foliation in $\mathbb{K}^{n+1}$ if it satisfies the *Frobenius integrability condition* given by the equation
$$\omega \wedge d\omega = 0,$$
see [4, Definition 2.2, p. 823].

(c) Let ω be an integrable 1-form and $L_X\omega$ the Lie derivative of ω with respect to a vector field X. Then, the solutions of the equation
$$L_X\omega = 0$$
define all the *infinitesimal automorphisms* of the foliation given by ω, see [4, Proposition 7.7, p. 845].

(d) Let ω be an integrable 1-form. Then the *tangent space of the space of foliations* at ω is given by the differential 1-forms η that satisfy the equation
$$\omega \wedge d\eta + d\omega \wedge \eta = 0,$$

2010 *Mathematics Subject Classification.* 53-04; 14-04.

Key words and phrases. Differential operators, Exterior algebra, Vector Fields.

The authors were supported by CONICET, Argentina.

see [2, Section 2.1. p. 709].

(e) Let D be a *bracket generating distribution*. Some important invariants of D are the ranks of the derived sequence

$$a(p) := \operatorname{rank} D^{(p)} = \operatorname{rank}\left(D^{(p-1)} + [D, D^{(p-1)}]\right),$$

see [5, §1, pp. 8-9].

(f) A *symplectic structure* in a variety of dimension $2r$ is given by a 2-form ω such that $d\omega = 0$ and $\omega^r \neq 0$, see [1, p. 41].

For a clear understanding on how `DiffAlg` deals with such equations, let us fix some notation.

Let $S = \mathbb{K}[x_0, \ldots, x_n]$ be the polynomial ring in $n+1$ variables and let $\Omega = \bigoplus_{r\geq 0} \Omega^r$ be the exterior algebra of differential forms of S over $\mathbb{K}$. Denote $\Omega^r(d)$ as the space of r-forms with polynomial coefficients of homogeneous degree d.

Therefore $\omega \in \Omega^r(d)$ can be written as

$$\omega = \sum_{\substack{I\subset\{0,\ldots,n\}\\ \#I=r}} \sum_{\substack{\alpha\in\mathbb{N}_0^{n+1}\\ |\alpha|=d}} a_{\alpha,I}\; x^\alpha\; dx_I, \qquad a_{\alpha,I} \in \mathbb{K}, \tag{1}$$

where each $I = \{i_1, \ldots, i_r\} \subset \{0, \ldots, n\}$ denote $dx_I := dx_{i_1} \wedge \ldots \wedge dx_{i_r}$ and each $\alpha = (\alpha_0, \ldots, \alpha_n) \in \mathbb{N}_0^{n+1}$ denote $|\alpha| := \sum_{i=0}^n \alpha_i$ and $x^\alpha := x_0^{\alpha_0} \cdot \ldots \cdot x_n^{\alpha_n}$.

Let T be the module of vector fields with coefficients in S. Analogously to eq. (1), $X \in T(e)$ can be written as

$$X = \sum_{i=0}^{n} \sum_{\substack{\beta\in\mathbb{N}_0^{n+1}\\ |\beta|=e}} b_{\beta,i}\; x^\beta\; \frac{\partial}{\partial x_i}, \qquad b_{\beta,i} \in \mathbb{K}.$$

Current algebraic software systems implement functionality to deal with differential forms and vector fields, but usually scalar parameters $a_{\alpha,I}$ and $b_{\beta,i}$ must be specified as fixed elements in $\mathbb{K}$. Instead, `DiffAlg` treats homogeneous forms and vector fields in a completely symbolic environment by considering the scalar coefficient rings

$$\mathbb{K}[a_{\alpha,I}] \qquad \text{and} \qquad \mathbb{K}[b_{\beta,i}].$$

Scalar coefficients can be systematically obtained by looking at the coordinates of differential forms and vector fields written in the standard basis

$$\mathcal{B}_{r,d} = \{x^\alpha dx_I\}_{\substack{\#I=r\\ |\alpha|=d}} \qquad \text{and} \qquad \mathcal{B}_e = \left\{x^\beta \tfrac{\partial}{\partial x_i}\right\}_{|\beta|=e}$$

of the spaces $\Omega^r(d)$ and $T(e)$, respectively.

Importantly, when using `DiffAlg`, each object is expected to be defined in its own coefficient ring. Then, computing some operations (wedge product, contraction, etc.) will involve different input and output spaces, falling back in a constant modification of the coefficients rings $\mathbb{K}[a_{\alpha,I}]$ or $\mathbb{K}[b_{\beta,i}]$. For greater clarity, consider the following example. Fix $\omega \in \Omega^r(d)$ and $X \in T(e)$ and consider the contraction $i_X\omega \in \Omega^{r-1}(d+e)$. Then, the following elements will be taking place:

Operation	(ω, X)	$\longmapsto$	$i_X\omega$
Rings	$\mathbb{K}[a_{\alpha,I}] \times \mathbb{K}[b_{\beta,i}]$		$\mathbb{K}[a_{\alpha,I}, b_{\beta,i}]$
Basis	$\mathcal{B}_{r,d} \times \mathcal{B}_e$		$\mathcal{B}_{r-1,d+e}$

As mentioned before, the main purpose of `DiffAlg` is to find algebraic solutions to equations in the context of differential algebra. Equations are treated differently in the linear and non-linear cases:

i) In the linear case, for example $i_R\omega = 0$, `DiffAlg` can compute a basis of the solutions of the equation. Once this is done, it can also compute a generic linear combination of the elements of the basis, see Example 1.
ii) In the non-linear case, for example $\omega \wedge d\omega = 0$, the coordinates will be polynomial. In this case `DiffAlg` would compute the ideal generating the space of solutions. This ideal can be obtained in two different ways: taking coordinates in the basis $\mathcal{B}_{r,d}$ or $\mathcal{B}_e$ or taking coordinates in the basis $\{dx_I\}$ or $\{\frac{\partial}{\partial x_i}\}$, see Examples 2 and 4.

`DiffAlg` can also be a valuable tool for studying differential operators. The lack of an algebraic theory to deal with such objects can be mitigated by non-conclusive computations easily made by `DiffAlg`. As a first example, one could consider computing solutions of the annihilation of a differential operator degree-wise for low degrees.

2. Some examples

Example 1. In the following example we obtain a basis of the space of projective differential 2-forms in $\mathbb{P}^3_{\mathbb{K}}$. Then, we define a generic projective differential 2-form to be possibly used in further computations.

```
i1 : loadPackage "DiffAlg";

i2 : R = radial 3;            -- create radial field in 4 variables

i3 : w = newForm(3,2,1,"a");   -- create a linear 2-form

i4 : K = genKer (R _ w, w);    -- get the forms that descend to projective space

i5 : length K       -- return the dimension of \Omega^2(1) in projective 3-space

o5 = 4

i6 : v = linearComb(K,"a")     -- define a generic projective form

o6 = (a x  - a x )dx dx  + (- a x  + a x )dx dx  + (a x  + a x )dx dx  +
       0 2    1 3   0  1       0 1    2 3   0  2     0 0    3 3   1  2
     ------------------------------------------------------------------------
     (a x  - a x )dx dx  + (-a x  - a x )dx dx  + (a x  + a x )dx dx
       1 1    2 2   0  3     1 0    3 2   1  3     2 0    3 1   2  3

o6 : DiffAlgForm
```

Example 2. In the finite dimensional $\mathbb{K}$-vector space $\Omega^1(d)$, the solutions of the equation $\omega \wedge d\omega = 0$ determine an algebraic variety, its points are the integrable differential 1-forms of degree d. In the following example, we compute the equations of the variety of integrable 1-forms of degree 1 in 3-dimensional space.

It is worth mention that, for $n \geq 3$ and $d > 5$, it is an open problem to classify the irreducible components of this varieties, see [2].

```
i1 : loadPackage "DiffAlg";

i2 : w = newForm (2,1,1,"a")    -- create a generic linear 1-form

o2 = (a x  + a x  + a x )dx  + (a x  + a x  + a x )dx  + (a x  + a x  + a x )dx
       0 0    3 1    6 2   0     1 0    4 1    7 2   1     2 0    5 1    8 2   2

o2 : DiffAlgForm

i3 : moduliIdeal (w ^ (diff w))    -- return the ideal given by w ^ (d w) = 0

o3 = ideal (- a a  + a a  + a a  - a a , - a a  + a a  + a a  - a a ,
               2 3    0 5    1 6    0 7     2 4    1 5    4 6    3 7
     ------------------------------------------------------------------------
            a a  - a a  + a a  - a a )
             5 6    2 7    1 8    3 8

               QQ[i]
o3 : Ideal of ------[][a , a , a , a , a , a , a , a , a ]
               2       0   1   2   3   4   5   6   7   8
              i  + 1
```

Example 3. Let D be a 2-dimensional distribution generated by vector fields X and Y in 5-dimensional space. In the following example we compute the ranks of the derived distributions $D^{(p)}$. We verify that D is bracket-generating, *e.g.*, its derived series eventually span the entire tangent space.

```
i2 : X = newField("x_0^2*ax_0+ x_1^2*ax_1+ x_2^2*ax_2+ x_3^2*ax_3");

i3 : Y = newField("x_5*ax_0+ x_4*ax_1+ x_3*ax_2+ x_2*ax_3+ x_1*ax_4+ x_0*ax_5");

i4 : D_0 = {X,Y};

i5 : for b in 1..3 do
     (for a in D_(b-1) do
     (D_b=join(D_(b-1),{a|Y,a|X}))); --compute the derived sequence

i6 : {rank dist D_0, rank dist D_1, rank dist D_2, rank dist D_3}

o6 = {2, 3, 5, 6}                   -- ranks of the derived series

o6 : List
```

Example 4. In the following example, we generate a random rational 1-form of type (1,2) in $\mathbb{P}^2_{\mathbb{K}}$. First, we compute (the dimension of) the space of its integrating factors, see [4, pp. 828-829]. Then, we compute the ideal of the singular locus of ω, *e.g.*, the ideal where ω vanishes.

```
i2 : w = random logarithmicForm (2,{1,2},"a",Projective => true);

i3 : f = newForm (2,0,3,"a");

i4 : length genKer(w^(diff f)+f * (diff w),f)

o4 = 2

i5 : I = singularIdeal w

                        2                          2    2
o6 = ideal (- 9x x  + 63x  - 36x x  - 54x x  - 54x , 9x  - 63x x  - 27x x  -
                0 1      1      0 2      1 2      2    0      0 1      0 2
     ------------------------------------------------------------------------
                 2     2               2
      45x x  - 54x , 36x  + 81x x  + 45x  + 54x x  + 54x x )
         1 2      2     0      0 1      1      0 2      1 2

               QQ[i]
o6 : Ideal of ------[][x , x , x ]
                2       0   1   2
              i  + 1
```

References

[1] R. L. Bryant, S. S. Chern, R. B. Gardner, H. L. Goldschmidt, and P. A. Griffiths, *Exterior differential systems*, Mathematical Sciences Research Institute Publications, vol. 18, Springer-Verlag, New York, 1991. MR 1083148 (92h:58007)

[2] F. Cukierman, J. V. Pereira, and I. Vainsencher, *Stability of foliations induced by rational maps*, Ann. Fac. Sci. Toulouse Math. (6) **18** (2009), no. 4, 685–715. MR 2590385 (2011d:32049)

[3] R. Hartshorne, *Algebraic geometry*, Springer-Verlag, New York, 1977, Graduate Texts in Mathematics, No. 52. MR 0463157 (57 #3116)

[4] T. Suwa, *Unfoldings of codimension one complex analytic foliation singularities*, Singularity theory (Trieste, 1991), World Sci. Publ., River Edge, NJ, 1995, pp. 817–865. MR 1378432 (97a:32041)

[5] N. Tanaka, *On differential systems, graded Lie algebras and pseudogroups*, J. Math. Kyoto Univ. **10** (1970), 1–82. MR 0266258 (42 #1165)

Departamento de Computación, Pabellón I, Ciudad Universitaria, CP C1428EGA, Buenos Aires, Argentina

E-mail address: manudubinsky@gmail.com

Departamento de Matemática, Pabellón I, Ciudad Universitaria, CP C1428EGA, Buenos Aires, Argentina

E-mail address: cmassri@dm.uba.ar

Departamento de Matemática, Pabellón I, Ciudad Universitaria, CP C1428EGA, Buenos Aires, Argentina

E-mail address: amoli@dm.uba.ar

Departamento de Matemática, Pabellón I, Ciudad Universitaria, CP C1428EGA, Buenos Aires, Argentina

E-mail address: fquallb@dm.uba.ar